Анимация без DirectX

При попытке изобразить некую анимацию использую только средства TCanvas, на экране получается черте-чего. Все мельтешит, дергается, одним словом - не годится. Для получения "гладкой" (не мельтешащей) анимация в программах не использующих DirectX, я обычно использую следующую технику.
Узким местом в процессе является момент изменения картинки на экране, поэтому рисование нужно проводить на невидимом для пользователя канвасе, и только подготовив там обновляемые участки выводить их на видимый экран. Для того, чтобы стереть картинку в том месте где ее уже нет, нужно помнить позицию в которой она была выведена в прошлый раз. Обзовем эту позицию Old: TRect, текущую позицию запомним в New: TRect. TRect я использую, на случай если размер отображаемой картинки может изменяться.
Стандартным подходом является написание двух процедур - Hide и Show, одна из которых прячет картинку в старой позиции, выводя участок фона поверх нее, а вторая выводит в новой позиции. Такой вариант не проходит и приводит к мерцанию изображения. Я предлагаю оставить процедуру Hide в покое, и пользоваться ей только если картинку нужно совсем убрать с экрана.
Процедура Show будет выполнять обе нужные функции. Для обновления экрана нам нужно погасить картинку в старой позиции и показать в новой. Тут возможны два варианта.
Первый - старый и новый прямоугольники пересекаются. В этом случае мы создаем временный TBItmap - tmp с размером их объединения, заполняем его требуемым участком фона, и рисуем на нем картинку. После такой подготовки выводим tmp в нужной позиции экрана.
Второй - старый и новый прямоугольники не пересекаются. В этом случае мы просто копируем прямоугольник old с невидимой копии фона на экран (процедура Hide), и рисуем нужную картинку в прямоугольнике new. При таком подходе мы избегаем двойной перерисовки экрана, что исключает мерцание.
Ниже программа которая все это делает.
var wsrf: TPaintBox; // видимый экран
var ssrf: TBitmap; // скрытый неизменяемый фон
var bmp : TBitmap; // картинка для анимации
var tmp : TBitmap; // временное хранилище
// пересекаются ли прямоугольники
function hasIntersect(const A, B: TRect): Boolean;
var
 R: TRect;
begin
 Result := False;
 R.Left := Max(A.Left, B.Left);
 R.Right := Min(A.Right, B.Right);
 if R.Left >= R.Right then Exit;
 R.Top := Max(A.Top, B.Top);
 R.Bottom := Min(A.Bottom, B.Bottom);
 if R.Top >= R.Bottom then Exit;
 Result := True;
end;
// результат - объединение
function Union(A, B: TRect): TRect;
begin
 if EmptyRect(A) then
 Result := B
 else if EmptyRect(B) then
 Result := A
 else
 begin
 Result.Left := Min(A.Left, B.Left );
 Result.Top := Min(A.Top, B.Top );
 Result.Right := Max(A.Right, B.Right );
 Result.Bottom := Max(A.Bottom, B.Bottom);
 end;
end;
procedure TOneTooth.Hide;
begin
 tmp.Width := bmp.Width;
 tmp.Height := bmp.Height;
 tmp.Canvas.CopyRect(bmpRect(tmp), ssrf.Canvas, old);
 wsrf.Canvas.Draw(old.Left, old.Top, tmp);
end;
procedure TOneTooth.Show;
var
 R, R1: TRect;
begin
 now.Right := now.Left + bmp.Width; // корректировка now на случай
 now.Bottom := now.Top + bmp.Height; // изменения размеров bmp
 if hasIntersect(old, now) then
 begin
 R := Union(old, now);
 tmp.Width := R.Right - R.Left;
 tmp.Height := R.Bottom - R.Top;
 tmp.Canvas.CopyRect(bmpRect(tmp), ssrf.Canvas, R); // фон
 tmp.Canvas.Draw(now.Left - r.Left, now.Top - r.Top, bmp); // фон + картинка
 end
 else
 begin
 Hide;
 tmp.Canvas.CopyRect(bmpRect(bmp), ssrf.Canvas, now); // фон
 tmp.Canvas.Draw(0, 0, bmp); // фон + картинка
 R := now;
 end;
 wsrf.Canvas.Draw(R.Left, R.Top, tmp);
 old := now;
end;
Author: http://sunsb.dax.ru 
Source: http://delphiworld.narod.ru
ID: 03808
Category:Графика